Topics of this Slideset

- Intro to Assembly language
- Programmer visible state
- Y86 Rudiments
- RISC vs. CISC architectures
Assembly Language View

- Processor state: registers, memory, etc.
- Instructions and how instructions are encoded

Layer of Abstraction

- Above: how to program machine, processor executes instructions sequentially
- Below: What needs to be built
  - Use variety of tricks to make it run faster
  - E.g., execute multiple instructions simultaneously
The Y86 is a “toy” machine that is similar to the x86 but much simpler. It is a gentler introduction to assembly level programming than the x86.

- just a few instructions as opposed to hundreds for the x86;
- fewer addressing modes;
- simpler system state;
- absolute addressing.

Everything you learn about the Y86 will apply to the x86 with very little modification. But the main reason we’re bothering with the Y86 is because we’ll be explaining pipelining in that context.
There are various means of giving a *semantics* or meaning to a programming system.

Probably the most sensible for an assembly (or machine) language is an *operational semantics*, also known as an *interpreter semantics*.

That is, we explain the semantics of each possible operation in the language by explaining the effect that execution of the operation has on the *machine state*. 
The most fundamental abstraction for the machine semantics for the x86/Y86 or similar machines is the **fetch-decode-execute** cycle.

The machine repeats the following steps forever:

1. **fetch** the next instruction from memory (the PC tells you which is next);
2. **decode** the instruction (in the control unit);
3. **execute** the instruction, updating the state appropriately;
4. go to step 1.
Program registers: almost the same as x86-64, each 64-bits
Condition flags: 1-bit flags set by arithmetic and logical operations. OF: Overflow, ZF: Zero, SF: Negative
Program counter: indicates address of instruction
Memory
  Byte-addressable storage array
  Words stored in little-endian byte order
Status code: (status can be AOK, HLT, INS, ADR) to indicate state of program execution.
We’re actually describing two languages: the assembly language and the machine language. There is nearly a 1-1 correspondence between them.

**Machine Language Instructions**

- 1-10 bytes of information read from memory
  - Can determine instruction length from first byte
  - Not as many instruction types and simpler encoding than x86-64
- Each instruction accesses and modifies some part(s) of the program state.
## Y86 Instruction Set

<table>
<thead>
<tr>
<th>Byte</th>
<th>0</th>
<th>1</th>
<th>2</th>
<th>3</th>
<th>4</th>
<th>5</th>
<th>6</th>
<th>7</th>
<th>8</th>
<th>9</th>
</tr>
</thead>
<tbody>
<tr>
<td>halt</td>
<td>0</td>
<td>0</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>nop</td>
<td>1</td>
<td>0</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>cmovXX rA,rB</td>
<td>2</td>
<td>fn</td>
<td>rA</td>
<td>rB</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>irmovq V,rB</td>
<td>3</td>
<td>0</td>
<td>F</td>
<td>rB</td>
<td></td>
<td>V</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>rmmovq rA,D(rB)</td>
<td>4</td>
<td>0</td>
<td>rA</td>
<td>rB</td>
<td>D</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>mrmovq D(rB),rA</td>
<td>5</td>
<td>0</td>
<td>rA</td>
<td>rB</td>
<td>D</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>OPq rA,rB</td>
<td>6</td>
<td>fn</td>
<td>rA</td>
<td>rB</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>jXX Dest</td>
<td>7</td>
<td>fn</td>
<td>Dest</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>call Dest</td>
<td>8</td>
<td>0</td>
<td>Dest</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>ret</td>
<td>9</td>
<td>0</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>pushq rA</td>
<td>A</td>
<td>0</td>
<td>rA</td>
<td>F</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>popq rA</td>
<td>B</td>
<td>0</td>
<td>rA</td>
<td>F</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
</tbody>
</table>
Each register has an associated 4-bit id:

<table>
<thead>
<tr>
<th>Register</th>
<th>ID</th>
<th>Function</th>
</tr>
</thead>
<tbody>
<tr>
<td>%rax</td>
<td>0</td>
<td>%r8, 8</td>
</tr>
<tr>
<td>%rcx</td>
<td>1</td>
<td>%r9, 9</td>
</tr>
<tr>
<td>%rdx</td>
<td>2</td>
<td>%r10, A</td>
</tr>
<tr>
<td>%rbx</td>
<td>3</td>
<td>%r11, B</td>
</tr>
<tr>
<td>%rsp</td>
<td>4</td>
<td>%r12, C</td>
</tr>
<tr>
<td>%rbp</td>
<td>5</td>
<td>%r13, D</td>
</tr>
<tr>
<td>%rsi</td>
<td>6</td>
<td>%r14, E</td>
</tr>
<tr>
<td>%rdi</td>
<td>7</td>
<td>no reg, F</td>
</tr>
</tbody>
</table>

Almost the same encoding as in x86-64.

Most of these registers are general purpose; %rsp has special functionality.
Y86 Instruction Set (2)

cmovXX rA,rB

Encompasses:

rrmovq rA,rB | 2 0 | move from register to register

cmovle rA,rB | 2 1 | move if less or equal

cmovl rA,rB | 2 2 | move if less

cmove rA,rB | 2 3 | move if equal

cmovne rA,rB | 2 4 | move if not equal

cmovge rA,rB | 2 5 | move if greater or equal

cmovg rA,rB | 2 6 | move if greater
OPq rA,rB

Encompasses:

- addq rA,rB  \[6\ 0\] add
- subq rA,rB  \[6\ 1\] subtract
- andq rA,rB  \[6\ 2\] and
- xorq rA,rB  \[6\ 3\] exclusive or
Encompasses:

- `jXX Dest`: unconditional jump
- `jle Dest`: jump if less or equal
- `jl Dest`: jump if less
- `je Dest`: jump if equal
- `jne Dest`: jump if not equal
- `jge Dest`: jump if greater or equal
- `jg Dest`: jump if greater
Simple Addressing Modes

- **Immediate**: value
  
  `irmovq 0xab, %rbx`

- **Register**: Reg[R]
  
  `rrmovq %rcx, %rbx`

- **Normal (R)**: Mem[Reg[R]]
  
  - Register R specifies memory address.
  - This is often called *indirect* addressing.

  `mrmovq (%rcx), %rax`

- **Displacement D(R)**: Mem[Reg[R] + D]
  
  - Register R specifies start of memory region.
  - Constant displacement D specifies offset

  `mrmovq 8(%rcb), %rdx`
It’s important to understand how individual operations update the system state. *But that’s not enough!*

Much of the way the Y86/x86 operates is based on a set of *programming conventions*. Without them, you won’t understand how programs work, what the compiler generates, or how your code can interact with code written by others.
The following are conventions necessary to make programs interact:

- How do you pass arguments to a procedure?
- Where are variables (local, global, static) created?
- How does a procedure return a value?
- How do procedures preserve the state/data of the caller?

Some of these (e.g., the direction the stack grows) are reflected in specific machine operations; others are purely conventions.
Let’s write a fragment of Y86 assembly code. Our program swaps the 8-byte values starting in memory locations 0x0100 (value A) and 0x0200 (value B).

```
start:
xorq    %rax, %rax
mrmovq  0x100(%rax), %rbx
mrmovq  0x200(%rax), %rcx
rmmovq  %rcx, 0x100(%rax)
rmmovq  %rbx, 0x200(%rax)
halt
```

<table>
<thead>
<tr>
<th>Reg.</th>
<th>Use</th>
</tr>
</thead>
<tbody>
<tr>
<td>%rax</td>
<td>0</td>
</tr>
<tr>
<td>%rbx</td>
<td>A</td>
</tr>
<tr>
<td>%rcx</td>
<td>B</td>
</tr>
</tbody>
</table>
Now, we generate the machine code for our sample program. Assume that it is stored in memory starting at location 0x030. *I did this by hand, so check for errors!*

<table>
<thead>
<tr>
<th>Address</th>
<th>Value</th>
<th>Instruction</th>
</tr>
</thead>
<tbody>
<tr>
<td>0x030</td>
<td>6300</td>
<td># xorq %rax, %rax</td>
</tr>
<tr>
<td>0x032</td>
<td>50300001000000000000000000000000</td>
<td># mrmovq 0x100(%rax), %rbx</td>
</tr>
<tr>
<td>0x03c</td>
<td>50010002000000000000000000000000</td>
<td># mrmovq 0x200(%rax), %rcx</td>
</tr>
<tr>
<td>0x046</td>
<td>40100001000000000000000000000000</td>
<td># rmrmovq %rcx, 0x100(%rax)</td>
</tr>
<tr>
<td>0x050</td>
<td>40300002000000000000000000000000</td>
<td># rmrmovq %rbx, 0x200(%rax)</td>
</tr>
<tr>
<td>0x05a</td>
<td>00</td>
<td># halt</td>
</tr>
</tbody>
</table>

<table>
<thead>
<tr>
<th>Reg.</th>
<th>Use</th>
</tr>
</thead>
<tbody>
<tr>
<td>%rax</td>
<td>0</td>
</tr>
<tr>
<td>%rbx</td>
<td>A</td>
</tr>
<tr>
<td>%rcx</td>
<td>B</td>
</tr>
</tbody>
</table>
Example from C to Assembly

Suppose we have the following simple C program in file code.c.

```c
int sumInts(long int n)
{
    /* Add the integers from 1..n. */
    long int i;
    long int sum = 0;
    for ( i = 1; i <= n; i++ ) {
        sum += i;
    }
    return sum;
}
```

(We used long int to force usage of the 64-bit registers.) You can compile it using the following commands:

```
> gcc -O -S code.c
```
.file "code.c"
.text
.globl sumInts
.type sumInts, @function

sumInts:
.LFB0:
  .cfi_startproc
  testq %rdi, %rdi
  jle .L4
  movq $0, %rax
  movq $1, %rdx
.L3:
  addq %rdx, %rax
  addq $1, %rdx
  cmpq %rdx, %rdi
  jge .L3
  ret
.L4:
  movq $0, %rax
  ret
  .cfi_endproc
.LFE0:
  .size sumInts, .-sumInts
  .ident "GCC: (Ubuntu 4.8.4-2ubuntu1~14.04) 4.8.4"
  .section .note.gnu-stack, ""
This is a hand translation into Y86 assembler:

```
sumInts:
    andq  %rdi, %rdi # test  %rdi = n
    jle   .L4       # if <= 0, done
    irmovq $1, %rcx # constant 1
    irmovq $0, %rax # sum = 0
    irmovq $1, %rdx # i = 1

.L3:
    rrmovq %rdi, %rsi # temp = n
    addq  %rdx, %rax # sum += i
    addq  %rcx, %rdx # i += 1
    subq  %rdx, %rsi # temp -= i
    jge   .L3       # if >= 0, goto L3
    ret

.L4:
    irmovq $0, %rax # done
    ret
```

How does it get the argument? How does it return the value?
By convention, the first 6 parameters to any procedure are passed in order in 6 specific registers. Others are passed on the stack in reverse order.

Registers: First 6 arguments

<table>
<thead>
<tr>
<th>%rdi</th>
<th>%rsi</th>
<th>%rdx</th>
<th>%rcx</th>
<th>%r8</th>
<th>%r9</th>
</tr>
</thead>
</table>

Mnemonic to remember the order: “Diane’s silk dress cost $89.”

Return value

%rax
Addition Instruction

- Add value in register rA to that in register rB.
  - Store result in register rB
  - Note that Y86 only allows addition to be applied to register data.

E.g., `addq %rax, %rsi` is encoded as: 60 06. **Why?**

- Set condition codes based on the result.
- Two byte encoding:
  - First indicates instruction type.
  - Second gives source and destination registers.

What effects does `addq` have on the state?
You completely characterize an operation by saying how it changes the state.

What effects does `addq %rsi, %rdi` have on the state?
You completely characterize an operation by saying how it changes the state.

What effects does `addq %rsi, %rdi` have on the state?

1. Set contents of `%rdi` to the sum of the current contents of `%rsi` and `%rdi`.
2. Set condition codes based on the result of the sum.
   - OF: set iff the result causes an overflow
   - ZF: set iff the result is zero
   - SF: set iff the result is negative
3. Increment the program counter by 2. Why 2?

There are no effect on the memory or status flag.
Arithmetic and Logical Operations

Add

- **addq rA, rB**
  - Encoding: 6 0
  - Registers: rA, rB

Subtract (rA from rB)

- **subq rA, rB**
  - Encoding: 6 1
  - Registers: rA, rB

And

- **andq rA, rB**
  - Encoding: 6 2
  - Registers: rA, rB

Exclusive Or

- **xorq rA, rB**
  - Encoding: 6 3
  - Registers: rA, rB

- Refer to generically as “OPq”
- Encodings differ only by “function code”: lower-order 4-bits in first instruction byte.
- Set condition codes as side effect.
### Move Operations

<table>
<thead>
<tr>
<th>Operation Type</th>
<th>Instruction</th>
<th>Opcode</th>
<th>Bit</th>
<th>Register A</th>
<th>Register B</th>
</tr>
</thead>
<tbody>
<tr>
<td><strong>Register to Register</strong></td>
<td><code>rrmovq rA, rB</code></td>
<td>2 0</td>
<td>rA</td>
<td>rB</td>
<td></td>
</tr>
<tr>
<td><strong>Immediate to Register</strong></td>
<td><code>irmovq V, rB</code></td>
<td>3 0</td>
<td>F</td>
<td>rB</td>
<td>V</td>
</tr>
<tr>
<td><strong>Register to Memory</strong></td>
<td><code>rmmovq rA, D(rB)</code></td>
<td>4 0</td>
<td>rA</td>
<td>rB</td>
<td>D</td>
</tr>
<tr>
<td><strong>Memory to Register</strong></td>
<td><code>mrmovq D(rB), rA</code></td>
<td>5 0</td>
<td>rA</td>
<td>rB</td>
<td>D</td>
</tr>
</tbody>
</table>

- Similar to the x86-64 movq instruction.
- Similar format for memory addresses.
- Slightly different names to distinguish them.
The Y86 adds special move instructions to compensate for the lack of certain *addressing modes*.
Move (conditionally)

```
cmovXX  rA,  rB  2 fn rA rB
```

- Refer to generically as “cmovXX”
- Encodings differ only by function code fn
- rrmovq instruction is a special case
- Based on values of condition codes
- Conditionally copy value from source to destination register
## Conditional Move Instructions

<table>
<thead>
<tr>
<th></th>
<th>2</th>
<th>0</th>
<th>rA</th>
<th>rB</th>
</tr>
</thead>
<tbody>
<tr>
<td><strong>Move Unconditionally</strong></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td><code>rrmovq rA, rB</code></td>
<td>2</td>
<td>0</td>
<td>rA</td>
<td>rB</td>
</tr>
</tbody>
</table>

<table>
<thead>
<tr>
<th></th>
<th>2</th>
<th>1</th>
<th>rA</th>
<th>rB</th>
</tr>
</thead>
<tbody>
<tr>
<td><strong>Move when less or equal</strong></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td><code>cmovle rA, rB</code></td>
<td>2</td>
<td>1</td>
<td>rA</td>
<td>rB</td>
</tr>
</tbody>
</table>

<table>
<thead>
<tr>
<th></th>
<th>2</th>
<th>2</th>
<th>rA</th>
<th>rB</th>
</tr>
</thead>
<tbody>
<tr>
<td><strong>Move when less</strong></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td><code>cmovl rA, rB</code></td>
<td>2</td>
<td>2</td>
<td>rA</td>
<td>rB</td>
</tr>
</tbody>
</table>

<table>
<thead>
<tr>
<th></th>
<th>2</th>
<th>3</th>
<th>rA</th>
<th>rB</th>
</tr>
</thead>
<tbody>
<tr>
<td><strong>Move when equal</strong></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td><code>cmove rA, rB</code></td>
<td>2</td>
<td>3</td>
<td>rA</td>
<td>rB</td>
</tr>
</tbody>
</table>

<table>
<thead>
<tr>
<th></th>
<th>2</th>
<th>4</th>
<th>rA</th>
<th>rB</th>
</tr>
</thead>
<tbody>
<tr>
<td><strong>Move when not equal</strong></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td><code>cmovne rA, rB</code></td>
<td>2</td>
<td>4</td>
<td>rA</td>
<td>rB</td>
</tr>
</tbody>
</table>

<table>
<thead>
<tr>
<th></th>
<th>2</th>
<th>5</th>
<th>rA</th>
<th>rB</th>
</tr>
</thead>
<tbody>
<tr>
<td><strong>Move when greater or equal</strong></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td><code>cmovge rA, rB</code></td>
<td>2</td>
<td>5</td>
<td>rA</td>
<td>rB</td>
</tr>
</tbody>
</table>

<table>
<thead>
<tr>
<th></th>
<th>2</th>
<th>6</th>
<th>rA</th>
<th>rB</th>
</tr>
</thead>
<tbody>
<tr>
<td><strong>Move when greater</strong></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td><code>cmovg rA, rB</code></td>
<td>2</td>
<td>6</td>
<td>rA</td>
<td>rB</td>
</tr>
</tbody>
</table>
Example of CMOV

Suppose you want to compile the following C code:

```c
long min (long x, long y) {
    if (x <= y)
        return x;
    else
        return y;
}
```

The following is one potential implementation of this. Notice that there are no jumps.

```assembly
min:
    rrmovq %rdi, %rax  # ans ← x
    rrmovq %rdi, %r8   # temp ← x
    subq %rsi, %r8    # if (temp − y) < 0
    cmovg %rsi, %rax  # ans ← y
    ret                # return ans
```
Jump Instructions

<table>
<thead>
<tr>
<th>jXX Dest</th>
<th>7</th>
<th>fn</th>
<th>Dest</th>
</tr>
</thead>
</table>

- Refer to generically as “jXX”
- Encodings differ only by function code fn
- Based on values of condition codes
- Same as x86-64 counterparts
- Encode full destination address (unlike PC-relative addressing in x86-64)
### Jump Instructions

#### Jump Unconditionally

<table>
<thead>
<tr>
<th>Instruction</th>
<th>Opcode</th>
<th>Flags</th>
<th>Comment</th>
</tr>
</thead>
<tbody>
<tr>
<td><code>jmp Dest</code></td>
<td>0x7</td>
<td>0x0</td>
<td>Dest</td>
</tr>
</tbody>
</table>

#### Jump when less or equal

<table>
<thead>
<tr>
<th>Instruction</th>
<th>Opcode</th>
<th>Flags</th>
<th>Comment</th>
</tr>
</thead>
<tbody>
<tr>
<td><code>jle Dest</code></td>
<td>0x7</td>
<td>0x1</td>
<td>Dest</td>
</tr>
</tbody>
</table>

#### Jump when less

<table>
<thead>
<tr>
<th>Instruction</th>
<th>Opcode</th>
<th>Flags</th>
<th>Comment</th>
</tr>
</thead>
<tbody>
<tr>
<td><code>jl Dest</code></td>
<td>0x7</td>
<td>0x2</td>
<td>Dest</td>
</tr>
</tbody>
</table>

#### Jump when equal

<table>
<thead>
<tr>
<th>Instruction</th>
<th>Opcode</th>
<th>Flags</th>
<th>Comment</th>
</tr>
</thead>
<tbody>
<tr>
<td><code>je Dest</code></td>
<td>0x7</td>
<td>0x3</td>
<td>Dest</td>
</tr>
</tbody>
</table>

#### Jump when not equal

<table>
<thead>
<tr>
<th>Instruction</th>
<th>Opcode</th>
<th>Flags</th>
<th>Comment</th>
</tr>
</thead>
<tbody>
<tr>
<td><code>jne Dest</code></td>
<td>0x7</td>
<td>0x4</td>
<td>Dest</td>
</tr>
</tbody>
</table>

#### Jump when greater or equal

<table>
<thead>
<tr>
<th>Instruction</th>
<th>Opcode</th>
<th>Flags</th>
<th>Comment</th>
</tr>
</thead>
<tbody>
<tr>
<td><code>jge Dest</code></td>
<td>0x7</td>
<td>0x5</td>
<td>Dest</td>
</tr>
</tbody>
</table>

#### Jump when greater

<table>
<thead>
<tr>
<th>Instruction</th>
<th>Opcode</th>
<th>Flags</th>
<th>Comment</th>
</tr>
</thead>
<tbody>
<tr>
<td><code>jg Dest</code></td>
<td>0x7</td>
<td>0x6</td>
<td>Dest</td>
</tr>
</tbody>
</table>
Suppose you want to count the number of elements in a null terminated list A with starting address in %rdi.

```
len:
  irmovq $0, %rax    # result = 0
  mrmovq (%rdi), %rdx # val = *A
  andq %rdx, %rdx    # Test val
  je Done            # If 0, goto
  Done               #  Done

Loop:
  .....               #
Done:
  ret
```
Region of memory holding program data.

Used in Y86 (and x86-64) for supporting procedure calls.

Stack top is indicated by `%rsp`, address of top stack element.

Stack grows toward lower addresses.

- Top element is at lowest address in the stack.
- When pushing, must first decrement stack pointer.
- When popping, increment stack pointer.
Stack Operations

**Push**

```
pushq rA  | a  0  rA  F
```
- Decrements `%rsp` by 8.
- Store quad word from `rA` to memory at `%rsp`.
- Similar to x86-64 pushq operation.

**Pop**

```
popq rA  | b  0  rA  F
```
- Read quad word from memory at `%rsp`.
- Save in `rA`.
- Increment `%rsp` by 8.
- Similar to x86-64 popq operation.
Subroutine call

<table>
<thead>
<tr>
<th>call</th>
<th>8</th>
<th>0</th>
<th>Dest</th>
</tr>
</thead>
</table>

- Push address of next instruction onto stack.
- Start executing instructions at Dest.
- Similar to x86-64 call instruction.

Subroutine return

| ret | 9 | 0 |

- Pop value from stack.
- Use as address for next instruction.
- Similar to x86-64 ret instruction.
No operation

| nop | 1 0 |

- Don’t do anything but advance PC.

Halt execution

| halt | 0 0 |

- Stop executing instructions; set status to HLT.
- x86-64 has a comparable instruction, but you can’t execute it in user mode.
- We will use it to stop the simulator.
- Encoding ensures that program hitting memory initialized to zero will halt.
## Status Conditions

<table>
<thead>
<tr>
<th>Mnemonic</th>
<th>Code</th>
<th>Meaning</th>
</tr>
</thead>
<tbody>
<tr>
<td>AOK</td>
<td>1</td>
<td>Normal operation</td>
</tr>
<tr>
<td>HLT</td>
<td>2</td>
<td>Halt inst. encountered</td>
</tr>
<tr>
<td>ADR</td>
<td>3</td>
<td>Bad address (instr. or data)</td>
</tr>
<tr>
<td>INS</td>
<td>4</td>
<td>Invalid instruction</td>
</tr>
</tbody>
</table>

**Desired behavior:**

- If AOK, keep executing
- Otherwise, stop program execution
Writing Y86 Code

Try to use the C compiler as much as possible.

- Write code in C.
- Compile for x86-64 with `gcc -Og -S`.
- Transliterate into Y86 code.
- Modern compilers make this more difficult, because they optimize by default.

To understand Y86 (or x86) code, you have to know the meaning of the statement, but also certain *programming conventions*, especially the *stack discipline*.

- How do you pass arguments to a procedure?
- Where are local variables created?
- How does a procedure return a value?
- How do procedures save and restore the state of the caller?
**Coding example:** Find number of elements in a null-terminated list.

```c
long len1( long a[] );
```

The answer in this case should be 3.
First try writing typical array code:

```c
/* Count elements in null-terminated list */
long len1( long a[] )
{
    long len;
    for ( len = 0; a[len]; len++ );
    return len;
}
```

**Problem:** Hard to do array indexing on Y86, since we don’t have scaled addressing modes.

**x86 Code:**

```
L3:
    addq $1, %rax
    cmpq $0, (%rdi,%rax,8)
    jne L3
```

Compile with `gcc -Og -S`
Second try: Write C code that mimics expected Y86 code.

```c
/* Count elements in null-terminated list */
long len2( long *a )
{
  long ip = (long) a;
  long val = *(long *) ip;
  long len = 0;
  while (val) {
    ip += sizeof(long);
    len++;
    val = *(long *) ip;
  }
  return len;
}
```

Result:
- Compiler generates exact same code as before!
- Compiler converts both versions into the same intermediate form.
Y86-64 Code Generation Example (3)

len:

```
irmovq $1, %r8  # Constant 1
irmovq $8, %r9  # Constant 8
irmovq $0, %rax  # len = 0
rmovq (%rdi), %rdx  # val = *a
andq %rdx, %rdx  # Test val
je Done  # If 0, goto Done
```

Loop:

```
addq %r8, %rax  # len++
addq %r9, %rdi  # a++
rmovq (%rdi), %rdx  # val = *a
andq %rdx, %rdx  # Test val
jne Loop  # If !0, goto Loop
```

Done:

```
ret
```

<table>
<thead>
<tr>
<th>Reg.</th>
<th>Use</th>
</tr>
</thead>
<tbody>
<tr>
<td>%rdi</td>
<td>a</td>
</tr>
<tr>
<td>%rax</td>
<td>len</td>
</tr>
<tr>
<td>%rdx</td>
<td>val</td>
</tr>
<tr>
<td>%r8</td>
<td>1</td>
</tr>
<tr>
<td>%r9</td>
<td>8</td>
</tr>
</tbody>
</table>
Program starts at address 0
- Must set up stack
  - Where located
  - Pointer values
  - Mustn’t overwrite data
- Must initialize data
init:
  # Set up stack pointer
  irmovq  Stack, %rsp
  # Execute main program
  call  Main
  # Terminate
  halt

# Array of 4 elements + final 0
  .align  8
Array:
  .quad  0x000d000d000d000d
  .quad  0x00c000c000c000c0
  .quad  0xb000b000b000b00
  .quad  0xa000a000a000a00
  .quad  0

- Program starts at address 0
- Must set up stack
- Must initialize data
- Can use symbolic names
Main:

irmovq Array, %r di

# call len(Array)
call len
ret

Set up call to len:

- Follow x86-64 procedure conventions
- Pass array address as argument
A program that translates Y86 code into machine language.

- 1-1 mapping of instructions to encodings.
- Resolves symbolic names.
- Translation is linear.
- Assembler directives give additional control.

Some common directives:

- `.pos x`: subsequent lines of code start at address $x$.
- `.align x`: align the next line to an $x$-byte boundary (e.g., long ints should be at a quadword address, divisible by 8).
- `.quad x`: put $x$ at the current address; a way to initialize a value.
Assembling Y86 Program

```
unix> yas len.ys
```

- Generates “object code” file `len.yo`
- Actually looks like disassembler output

<table>
<thead>
<tr>
<th>Address</th>
<th>Assembly</th>
</tr>
</thead>
<tbody>
<tr>
<td>0x054</td>
<td><code>len</code></td>
</tr>
<tr>
<td>0x054:</td>
<td><code>30f8010000000000000000000</code></td>
</tr>
<tr>
<td>0x05e:</td>
<td><code>30f9080000000000000000000</code></td>
</tr>
<tr>
<td>0x068:</td>
<td><code>30f0000000000000000000000</code></td>
</tr>
<tr>
<td>0x072:</td>
<td><code>502700000000000000000000</code></td>
</tr>
<tr>
<td>0x07c:</td>
<td><code>6222</code></td>
</tr>
<tr>
<td>0x07e:</td>
<td><code>73a000000000000000000000</code></td>
</tr>
<tr>
<td>0x087:</td>
<td><code>6080</code></td>
</tr>
<tr>
<td>0x089:</td>
<td><code>6097</code></td>
</tr>
<tr>
<td>0x08b:</td>
<td><code>502700000000000000000000</code></td>
</tr>
<tr>
<td>0x095:</td>
<td><code>6222</code></td>
</tr>
<tr>
<td>0x097:</td>
<td><code>748700000000000000000000</code></td>
</tr>
<tr>
<td>0x0a0:</td>
<td><code>90</code></td>
</tr>
<tr>
<td>0x0a0:</td>
<td><code>90</code></td>
</tr>
</tbody>
</table>

```
0x054 : irmovq $1 , %r8
0x054 : irmovq $8 , %r9
0x054 : irmovq $0 , %rax
0x05e : rmovq (%rdi) , %rdx
0x068 : irmovq $1 , %r8
0x068 : irmovq $8 , %r9
0x068 : irmovq $0 , %rax
0x072 : rmovq (%rdi) , %rdx
0x07c : andq %rdx , %rdx
0x07e : je Done
```

<table>
<thead>
<tr>
<th>Address</th>
<th>Assembly</th>
</tr>
</thead>
<tbody>
<tr>
<td>0x087 :</td>
<td><code>Loop</code></td>
</tr>
<tr>
<td>0x087 :</td>
<td><code>addq %r8 , %rax</code></td>
</tr>
<tr>
<td>0x089 :</td>
<td><code>addq %r9 , %rdi</code></td>
</tr>
<tr>
<td>0x08b :</td>
<td><code>rmovq (%rdi) , %rdx</code></td>
</tr>
<tr>
<td>0x095 :</td>
<td><code>andq %rdx , %rdx</code></td>
</tr>
<tr>
<td>0x097 :</td>
<td><code>jne Loop</code></td>
</tr>
<tr>
<td>0x0a0 :</td>
<td><code>Done : ret</code></td>
</tr>
</tbody>
</table>
Simulating Y86 Programs

unix> yis len.yo

Instruction set simulator
- Computes effect of each instruction on process state
- Prints changes in state from original

Stopped in 33 steps at PC = 0x13, Status 'HLT', CC Z=1
S=0 O=0

Changes to registers:
%rax: 0x0000000000000000 0x0000000000000004
%rsp: 0x0000000000000000 0x0000000000000100
%rdi: 0x0000000000000000 0x000000000000038
%r8: 0x0000000000000000 0x0000000000000001
%r9: 0x0000000000000000 0x0000000000000008

Changes to memory:
0x00f0: 0x0000000000000000 0x0000000000000053
0x00f8: 0x0000000000000000 0x0000000000000013
Complex Instruction Set Computer

- Dominant ISA style through the 80s.
- Lots of instructions:
  - Variable length
  - Stack as mechanism for supporting functions
  - Explicit push and pop instructions.
- ALU instructions can access memory.
  - E.g., `addq %rax, 12(%rbx, %rcx, 8)`
  - Requires memory read and write in one instruction execution.
  - Some ISAs had much more complex address calculations.
- Set condition codes as a side effect of other instructions.
- Basic philosophy:
  - Memory is expensive;
  - Instructions to support high-level language constructs.
Reduced Instruction Set Computer

- Originated in IBM Research; popularized in Berkeley and Stanford projects.
- Few, simple instructions.
  - Takes more instructions to execute a task, but faster and simpler implementation
  - Fixed length instructions for simpler decoding
- Register-oriented ISA
  - More registers (32 typically)
  - Stack is back-up for registers
- Only load and store instructions can access memory (mrmovq and rmmovq in Y86).
- Explicit test instructions set condition values in register.
- Philosophy: KISS
CISC vs. RISC

Original Debate
- Strong opinions!
- CISC proponents—easy for compiler, fewer code bytes
- RISC proponents—better for optimizing compilers, can make run fast with simple chip design

Current Status
- For desktop processors, choice of ISA not a technical issue
  - With enough hardware, can make anything run fast
  - Code compatibility more important
- x86-64 adopted many RISC features
  - More registers; use them for argument passing
- For embedded processors, RISC makes sense
  - Smaller, cheaper, less power
  - Most cell phones use ARM processor
**Y86-64 Instruction Set Architecture**
- Similar state and instructions to x86-64
- Simpler encodings
- Somewhere between CISC and RISC

**How Important is ISA Design?**
- Less now than before: with enough hardware, can make almost anything run fast!